
log using "D:\Jason\workshop\Categorical data analysis\2019\categorical.log", replace

***************************
* Chi-Square tests
***************************

webuse lbw, clear
des
sum

******************************************
* Creating a 2- to 4-way table
*******************************************

* a 2-way table

table low ui, contents(freq)
tab low ui

* a 3-way table.

table low ui ht, contents(freq)

* alternative commands for a 3-way table

sort ht
by ht: tab low ui

* a 4-way table.

table low ui ht, by(race)  contents(freq)

* alterntive commands for a 4-way table
sort race ht
by race ht: tab low ui

*********************************************************
*  Test the Chi-Square association for a two-way table
**********************************************************

tab2 low ui, mis chi lrchi2 exac

**************************************************
*  Test the Chi-Square association for a three-way table
***************************************************

* testing the main effect of three variables only
ipf, fit(low+ui+ht)

*testing the main effects of three variables 
* and one interaction term between low and ui
ipf, fit (low+ui+ht+ low*ui)

*************************
* Logistic Regression
*************************

webuse lbw, clear
des
list in 1/20

* look at the regression coefficients
logit low age lwt i.race smoke ptl ht ui

* Look at the odds ratio
logit low age lwt i.race smoke ptl ht ui,or

* look at the predicted probability

margins i.race, atmeans

***********************
* Ordered Logistic Regression
***********************

use "http://fmwww.bc.edu/repec/bocode/g/gss.dta", clear
recode degree 4=3
label define degree 0 "lt high school" 1 "high school"  ///



2 "junior college" 3 "college", modify
label value degree degre
save d:\temp\order.dta, replace

use d:\temp\order.dta, clear
des
list in 1/20

* look at the regression coefficients
ologit degree south c.coh##c.coh i.black paeduc

* Look at the odds ratio
ologit degree south c.coh##c.coh i.black paeduc, or

* look at the predicted probability
margins i.black, atmeans

**************************************
* Sequential logistic Regression
**************************************

   use d:\temp\order.dta, clear
    des
    list in 1/20        
    seqlogit degree south c.coh##c.coh if black == 0 ,     ///
    tree(0 : 1 2 3 , 1 : 2 3 , 2 : 3 )                     ///
    ofinterest(paeduc)  over(c.coh##c.coh)                 ///
    levels(0=9, 1=12, 2=14, 3=16)

    seqlogitdecomp, overat(coh 1.5,       ///
                           coh 2.5,       ///
                           coh 3.5,       ///
                           coh 4.5,       ///
                           coh 5.5,       ///
                           coh 6.5)       ///
       at(south 0 paeduc 12)              ///
       yline(0) xline(0)                  ///
       subtitle("1915" "1925" "1935"      ///
                "1945" "1955" "1965")     ///
       eqlabel(`""less than high school" "versus" "high school or more""' ///
               `""high school" "versus" "any college""'                   ///
               `""junior college" "versus" "college""' )

**********************
* Multi-nomial Logistic Regression
***********************

use http://www.stata-press.com/data/r13/choice, clear
keep if choice == 1
des
list in 1/30

* obtain regression coefficients
mlogit car i.sex income

* obtain relative-risk ratios,
mlogit car i.sex income, rrr

* obtain predicted probability
margins sex, atmeans
margins, over(sex) dydx(income)

************************************
* Conditional Logistic Regression
************************************

use http://www.stata-press.com/data/r13/union, clear
des
sort idcode
list in 1/20, sepby(idcode)

clogit union age grade not_smsa south black, group(idcode)

***************************



* Nested Logistic Regression
***************************

webuse restaurant

des

tab1 restaurant, mis

sort family_id restaurant

list in 1/30, sepby(family_id)

nlogitgen type = restaurant(fast: Freebirds | MamasPizza, family:  CafeEccell | LosNortenos | WingsNmore, fancy: Christo

nlogittree restaurant type, choice(chosen) case(family_id)

nlogit chosen cost distance rating || type: income kids, base(family) || restaurant:, noconst case(family_id)

********************************************
* Alternative-Specific Conditional Logit 
*******************************************
use http://www.stata-press.com/data/r13/choice, clear

sort id car
list in 1/30, sepby(id)
asclogit choice dealer, case(id) alternatives(car) casevars(sex income)

log close


